

**JUEGOS DE ORDENADOR**

# **TERROR**

**... COMMODORE 64 ... VIC 20 ... APPLE ... SPECTRUM ... DRAGÓN ... MSX ...**



**EDICIONES GENERALES ANAYA**



# TERROR

Jenny Tyler y Chris Oxlade

Ediciones Generales Anaya



## Contenido

- 3 Pesadilla informática
- 4 Los números mágicos
- 6 El atrapa-fantasmas
- 8 La mujer-araña
- 10 El enterrador
- 12 La casa fantasma
- 14 El laberinto  
fantasmagórico
- 16 Espiritismo
- 17 Respuestas a las  
cuestiones planteadas

# Acerca de este libro

Los programas de este libro están escritos en una versión estándar de BASIC, uno de los lenguajes de ordenador, y hay líneas de conversión para la mayoría de los principales tipos de ordenadores personales. Busca a la izquierda del listado del programa el símbolo que representa a tu ordenador y después mira en la lista de cambios los necesarios para la versión correcta de esa línea. Los símbolos para los distintos ordenadores son los siguientes:

- ▲ Commodore 64 y VIC 20
- Spectrum
- Apple
- ★ DRAGON
- MSX

## Acerca de los juegos

Los juegos de este libro son muy sencillos. Se han creado con la intención de ayudarte a acostumbrarte a tu ordenador y al lenguaje BASIC, tecleando los programas, depurándolos y comprendiendo cómo funcionan. Los programas no tienen gráficos o sonido porque varían mucho de un ordenador a otro, pero puedes intentar añadirseles.

Puedes cambiar y adaptar los juegos a tu gusto. Detrás de cada listado de programa se sugieren maneras de hacerlo, pero también puedes experimentar con tus propias ideas. De esta manera, puedes usar los juegos de este libro como base para conseguir juegos más largos y más complicados hechos por ti mismo.

**Pesadilla informática** está basado en una idea de Brendon Kavanagh  
**Los números mágicos, El atrapa-fantasmas y El laberinto fantasmagórico** fueron escritos por Colin Reynolds  
**La Mujer-araña** está escrito por Val Robinson; **El enterrador**, por Alan Ramsey y **La Casa Fantasma**, por Keith Campbell

**Ilustrado por Rob McCaig**  
**Editado por Jenny Tyler**

*Queda prohibida la reproducción total o parcial de la presente obra bajo cualquiera de sus formas, gráficas o audiovisuales sin la autorización previa y escrita del editor, excepto citas en revistas, diarios o libros, siempre que se mencione la procedencia de las mismas.*

## Teclear y ejecutar los programas

Recuerda que puede resultar difícil y llevar mucho tiempo teclear correctamente los programas, incluso si éstos son cortos. Comprueba cada línea que introduzcas, ya que es fácil cometer errores aunque tengas bastante experiencia.

Cuando hayas introducido la totalidad del programa, compruébalo otra vez, asegurándote de que no te has saltado ninguna línea, espacios o signos de puntuación.

Para empezar el juego, teclea RUN. Antes de empezar, lee la introducción del juego para tener una idea de lo que se supone que éste debe hacer. Si el programa no funciona correctamente, es bastante probable que haya un error en cualquier sitio; por eso, lista el programa y compruébalo de nuevo.

Al acabar el juego, puede que el ordenador te pregunte si quieres jugar otra vez o, por el contrario, decir algo como BREAK in 200, en cuyo caso debes teclear RUN para jugar otra vez.

## Cambiar la velocidad

Algunos juegos dependen de la velocidad de tus reacciones y del ordenador. Puede ser necesario ajustar la velocidad. Encontrarás instrucciones para hacerlo a continuación del listado del programa.

Título original: Creepy Computer Games  
© 1983 Usborne Publishing Ltd.  
© 1985, de la edición española,  
Ediciones Generales Anaya  
Villafranca, 22. 28028 Madrid  
ISBN: 84-7525-311-3  
Depósito legal: M. 39.272/1985  
Impreso por Melsa  
Carretera de Fuenlabrada a Pinto, km. 21,800  
Pinto (Madrid)  
Impreso en España - Printed in Spain



# Pesadilla informática

Acostumbas a quedarte hasta tarde por la noche delante del ordenador y has caído dormido frente al teclado. Repentinamente, tu ordenador cobra vida y comienza a arrojar números y a abusar de ti. Para combatirlo, tienes que teclear los números según aparecen en la pantalla. Tu puntuación de comienzo es 300, y se incrementa si pulsas el número correcto y disminuye si no lo haces. Si consigues una puntuación por encima de 500, el ordenador cederá y tú ganas; pero si llega a 0, te convertirás en esclavo de tu ordenador. ¡Es una pesadilla real! ¿Puedes mantenerte a salvo?

## Cómo funciona

5 LET F\$=""	
10 DIM C\$(5)	
20 LET S=300	Fija el valor de comienzo del tanteo.
30 LET C\$(1)="** ORDENADORES AL PODER! **"	
40 LET C\$(2)="* LAS PERSONAS SOIS ESTUPIDAS *"	
50 LET C\$(3)=" UN ROBOT PARA PRESIDENTE!"	
60 LET C\$(4)=" LOS ORDENADORES SON LOS MEJORES!"	Almacena comentarios en la memoria.
70 LET C\$(5)="* YO SOY MEJOR QUE TU!*"	
80 CLS	Limpia la pantalla.
90 LET N=INT(RND(1)*9)+1	Elige un número entre 1 y 9 y lo coloca en N.
100 PRINT TAB(5);N	
110 PRINT TAB(15);S	Imprime N y el tanteo (S) en puntos diferentes de la pantalla.
120 IF RND(1)>.5 THEN GOTO 150	Decide si salta a 150.
130 PRINT	
140 PRINT C\$(INT(S/100)+1)	
150 IF S>60 THEN GOTO 160	
155 PRINT "< NO HAY ESPERANZA >"	
160 IF S>440 THEN PRINT " SOCORRO! AYUDA!!"	Imprime un mensaje, dependiendo del tanteo.
170 FOR I=1 TO 200	
180 LET A\$=INKEY\$	Comprueba si estás pulsando una tecla. Si es así, almacena su valor en FS.
190 IF A\$<>"**" THEN LET F\$=A\$	
200 NEXT I	
210 LET S=S-10	Disminuye en 10 el tanteo.
220 IF VAL(F\$)<>N THEN GOTO 240	Si pulsas una tecla equivocada, el ordenador salta a la línea 240 sin realizar el incremento de puntuación de la línea 230.
230 LET S=S+10*N*2	
240 IF S<0 THEN GOTO 270	
250 IF S>500 THEN GOTO 290	Comprueba el tanteo para ver si has ganado o perdido y si es así, salta al final del juego.
260 GOTO 80	Regresa para otra baza.
270 PRINT "AHORA ERES MI ESCLAVO"	
280 STOP	
290 PRINT "OK, TU GANAS (POR AHORA)"	
300 STOP	

## Líneas de conversión

5 LET F\$="0"	
10 DIM C\$(5,23)	
80 PRINT CHR\$(147)	
80 HOME	
90 LET N=INT(RND*9+1)	
90 LET N=INT(RND(0)*9+1)	
90 LET N=INT(RND(-TIME)*9+1)	
120 IF RND>0.5 THEN GOTO 150	
120 IF RND(0)>0.5 THEN GOTO 150	
120 IF RND(-TIME)>0.5 THEN GOTO 150	
170 FOR I=1 TO 100	
175 LET A\$=""	
180 GET A\$	
180 IF PEEK(-16384)>127 THEN GET A\$	

La velocidad del juego depende de este número. Probablemente necesites ajustarlo para adecuar la velocidad de tu ordenador a tus reacciones. Un número más alto hace el juego más lento.


## Prueba estos cambios

Las líneas 20, 210 y 230 controlan el tanteo. Cambia los números de estas líneas y observa qué ocurre.

## Más cosas

¿Puedes imaginar una manera de modificar el programa para que imprima letras en la pantalla, en vez de números?

# Los números mágicos



Te enfrentas ahora al Mago de los Números. El hechicero conjura en el aire los números de 1 a 9 y tira dos dados. Debes darle dos números que estén en el aire en ese momento y que sumados den el mismo número que suman los dos números de los dados. Una vez has usado un número, su imagen desaparece del aire y no puedes usarlo otra vez. Si han desaparecido todos los números antes de usar todas tus bazas has ganado. (Consigues una baza extra cada vez que el hechicero saca un doble.) Puedes usar ceros tantas veces como quieras en alguno de tus números. Si no puedes jugar, pasa a la baza siguientes usando dos ceros. Comprueba cuántas veces puedes ganar al hechicero.

## Cómo funciona el programa

```
10 PRINT "LOS NUMEROS MAGICOS"
20 DIM A(10)
```

Dimensiona el array, A.

```
30 LET T=8
40 FOR I=1 TO 10
50 LET A(I)=1
60 NEXT I
70 CLS
80 LET A(1)=1
```

Fija el número de bazas permitidas.

Un bucle coloca un 1 en cada espacio de A.

Se asegura de que A(1) siempre contiene un 1 (es así porque puedes usar el 0 tantas veces como quieras).

V lleva la cuenta de cuántos números se han usado. (Puesto a cero al empezar.)

```
90 LET V=0
100 FOR I=2 TO 10
110 IF A(I)<>0 THEN GOTO 150
120 PRINT " ";
130 LET V=V+1
140 GOTO 160
150 PRINT I-1;
160 NEXT I
170 PRINT
180 IF V=9 THEN GOTO 370
```

El bucle imprime los números. Si A(I)=0, el número se ha usado ya; por eso se deja un espacio y se suma 1 a V.

Si se han usado todos los números, salta al final del juego.

```
190 PRINT "TE QUEDAN ";T;" INTENTOS"
```

Te dice cuántas bazas te quedan.

```
200 LET C=INT(RND(1)*6+1)
210 LET B=INT(RND(1)*6+1)
220 PRINT "EL VALOR DE LOS DADOS ES ";C;" ";B
230 PRINT " CUALES SON TUS NUMEROS ?"
240 IF B=C THEN LET T=T+2
```

Elige los números para la tirada de los dados y los imprime.

Si la tirada es un doble, te regala una baza extra.

```
250 LET T=T-1
260 IF T<=0 THEN GOTO 350
270 INPUT N
280 INPUT M
290 IF M>9 OR N>9 THEN PRINT
    "DEMASIADO GRANDE - PRUEBA OTRA VEZ"
300 IF M>9 OR N>9 THEN GOTO 230
```

Resta 1 a las bazas que quedan; comprueba que T no es cero.

Coge tus dos números y comprueba si son válidos.

```
310 IF M+N<>B+C OR A(N+1)=0 OR
    A(M+1)=0 THEN GOTO 70
```

Si la suma no es correcta o si alguno de ellos ya ha sido usado, regresa para jugar otra baza.

```
320 LET A(M+1)=0
330 LET A(N+1)=0
340 GOTO 70
350 PRINT "EL HECHICERO GANA"
360 STOP
370 PRINT "TU GANAS"
380 STOP
```

Si los números son correctos, se sustituyen los unos en el array A por ceros, para indicar que se han usado.

## Líneas de conversión

```
▲ 70 PRINT CHR$(147)
● 70 HOME
□ 200 LET C=INT(RND*6+1)
■ 200 LET C=INT(RND(0)*6+1)
★ 200 LET C=INT(RND(-TIME)*6+1)
□ 210 LET B=INT(RND*6+1)
■ 210 LET B=INT(RND(0)*6+1)
★ 210 LET B=INT(RND(-TIME)*6+1)
```

Quizá puedas encontrar una manera de hacer que el ordenador comience otro juego automáticamente y construir un registro de cuántas veces has ganado o perdido. También podrías imaginar un terrible destino para los jugadores que pierdan demasiadas veces...

# El atrapa-fantasmas



Los fantasmas de este juego son números que cruzan rápidamente la pantalla. Para atraparlos, activa tu máquina usando la tecla X; pero recuerda: sólo funcionará cuando su número sea el mismo que el del fantasma atacante. Puedes incrementar el número de la máquina pulsando la tecla M (cuando llega a 9, se pone a cero otra vez). Si no consigues cazar al fantasma, éste te quitará una vida (marcadas como una «/» en la parte superior izquierda de la pantalla). A ver qué tal eres cazando fantasmas.

## Cómo funciona el programa

10 PRINT "ATRAPA FANTASMAS"	
20 LET S=0	] _____
30 LET Y=0	
40 LET L=3	_____
<input type="checkbox"/> ★ 50 LET N=INT(RND(1)*10)	_____
60 LET I=1	_____
●▲ 70 CLS	] _____
80 FOR J=1 TO L	
90 PRINT "/";	
100 NEXT J	
110 PRINT	
120 PRINT TAB(I);N;TAB(18);";";Y	_____

Fija a cero el tanteo y tu número al comienzo del juego.

Establece 3 como número de «vidas».

Toma un número entre 0 y 9 para el fantasma.

Inicializa el contador para la distancia que el fantasma puede avanzar antes de alcanzarte.

Bucle que imprime tus «vidas» como «/» en la parte superior izquierda de la pantalla.

Imprime el número del fantasma, la barrera «:» y a ti en posiciones apropiadas en la pantalla.

Este es un buen juego para añadirle sonido. Intenta añadir subrutinas para algo como el sonido del borboteo de la sangre cuando pierdas una vida y algo que parezca el sonido que puede hacer un fantasma cuando lo cazas.



```

☐ 125 FOR K=1 TO 50
☐ 127 NEXT K
●▲ 130 LET B$=INKEY$
140 IF B$="X" THEN GOTO 220
150 IF B$="M" THEN LET Y=Y+1

160 IF Y=10 THEN Y=0

170 LET I=I+1
180 IF I<18 THEN GOTO 70

190 LET L=L-1

200 IF L>0 THEN GOTO 50
210 GOTO 270
220 IF Y<>N THEN GOTO 170
230 PRINT "ATRAPADO"
240 PRINT "*****"
245 FOR K=1 TO 300
247 NEXT K
250 LET S=S+(18-I)
260 GOTO 50
●▲ 270 CLS
280 PRINT "TU TANTEO ATRAPANDO"
290 PRINT "FANTASMAS ES ";S
300 PRINT
310 PRINT "OTRA PARTIDA? (PULSA S O N)"
320 INPUT A$
330 IF A$="S" THEN RUN
340 IF A$="N" THEN STOP

```

Puedes alterar la velocidad de este juego cambiando el número de la línea 125. Usa un número más bajo para hacerlo más rápido. También puedes hacer el juego más duro reduciendo el número de la línea 120.



Comprueba si has pulsado una tecla. Si es X (es decir, si has activado la máquina de cazar fantasmas), el ordenador salta a otro punto del programa para comprobar si te está permitido esto. Si es M, incrementa en 1 tu número.

Pone a cero tu número si llega a 10.

Incrementa el contador de la distancia al fantasma. Si no ha alcanzado la barrera y no lo has cazado, el ordenador vuelve a imprimir el fantasma en la siguiente posición.

Si alcanza la barrera, pierdes una vida.

Vuelve por otro fantasma si todavía tienes vidas.

Comprueba si tu número es igual al del fantasma cuando lo quieres cazar. Si no, regresa por otro fantasma.

### Líneas de conversión


```

☐ 50 LET N=INT(RND*10)
☒ 50 LET N=INT(RND(0)*10)
★ 50 LET N=INT(RND(-TIME)*10)
▲ 70 PRINT CHR$(147)
● 70 HOME
☐ borrar 125
☐ borrar 127
● 128 LET B$=""
▲ 130 GET B$
● 130 IF PEEK(-16384))127 THEN GET B$
▲ 270 PRINT CHR$(147)
● 270 HOME

```

# La mujer-araña

¡Horror! ¡Es la mujer-araña!  
Estás de suerte, porque hoy está de buen humor. Si puedes adivinar rápidamente la letra que está pensando, te dejará libre. Si no, probablemente te convertirás en una mosca.  
Para averiguar la letra, dale una palabra y te dirá si su letra está o no en ella.



```
10 LET G=0
●▲ 20 CLS
```

## Cómo funciona

```

□■★ 30 LET T=INT(RND(1)*26+65)
40 LET T$=CHR$(T)
50 PRINT "LA MUJER ARANA"
60 PRINT "HA ELEGIDO"
70 PRINT
80 PRINT "PRUEBA UNA PALABRA"
90 PRINT
100 PRINT
110 INPUT W$
120 LET G=G+1
130 LET L=LEN(W$)
140 IF L<4 OR L>8 THEN GOTO 70
150 LET F=0
160 FOR I=1 TO L
170 LET A$=MID$(W$,I,1)
180 IF A$=T$ THEN LET F=1
190 NEXT I
200 IF F=1 THEN GOTO 280
210 PRINT
220 PRINT "NO ESTA EN ESA PALABRA"
230 FOR A=1 TO 500
240 NEXT A
250 IF G>15 THEN GOTO 400

●▲ 260 CLS
270 GOTO 70
280 PRINT "SI - ES UNA DE ESAS"
290 PRINT
300 PRINT "QUIERES INTENTAR ADIVINAR?"
305 PRINT "(S O N)"
310 INPUT R$
320 IF R$="N" THEN GOTO 250
330 PRINT
340 PRINT "CUAL ES TU LETRA?"
350 INPUT G$
360 IF G$<>T$ THEN GOTO 410
370 PRINT "ESTA BIEN - PUEDES IRTE"
380 PRINT "(DE MOMENTO)"
390 STOP
400 PRINT "DEMASIADO TARDE"
410 PRINT "AHORA ERES UNA MOSCA"
420 STOP

```

Elige un número, lo convierte en una letra y lo pone en T\$.

Pide una palabra.

Lleva la cuenta del número de intentos.

Comprueba si la palabra es demasiado larga o demasiado corta.

Analiza tu palabra para comprobar si la letra está en ella. Si lo está, va a 280.

Imprime el mensaje si la letra no se encuentra en esa palabra.

Bucle de demora para que puedas leer el mensaje.

Finaliza el programa si has hecho 15 intentos.

Limpia la pantalla. Vuelve por otra palabra.

Te hace saber si la letra se encuentra en tu palabra. Te pregunta si quieres intentar adivinarla. Si «no», vuelve por otra palabra.

Imprime el mensaje, dependiendo de si has adivinado la letra o no.

## Líneas de conversión

```

▲ 20 PRINT CHR$(147)
● 20 HOME
| 30 LET T=INT(RND*26+65)
■ 30 LET T=INT(RND(0)*26+65)
★ 30 LET T=INT(RND(-TIME)*26+65)
| 170 LET A$=W$(I TO I)
| 230 FOR A=1 TO 20
▲ 260 PRINT CHR$(147)
● 260 HOME

```

## Experimentos

Observa qué ocurre si cambias el 15 de la línea 250 por un número más pequeño. ¿Qué pasa si cambias los números 4 y 8 de la línea 140?

## Cambiar el juego

La mujer-araña está aburrida. Quiere que ajustes el programa para hacer lo siguiente:

Si su letra está en tu palabra, se te permiten dos intentos para adivinarla; pero si tu intento es equivocado pierdes 5 oportunidades. ¿Puedes hacerlo?

# El Enterrador

Es de noche y aúlla el viento; no es la noche ideal para perderse en un cementerio, pero allí es donde estás. Tienes hasta la medianoche para encontrar la salida. Los esqueletos acechan en las sombras, esperando para darte un susto de muerte cuando te acerques demasiado. Puedes cavar hoyos para conseguir que se mantengan lejos, pero hacerlo es un trabajo cansado y no puedes cavar más de 5 en un juego. Has de tener cuidado para no caer en los hoyos que tú mismo has cavado. Las lápidas (representadas por «+») y las vallas del cementerio (representadas por «:») bloquean tu camino. Los hoyos que cavas se representan por «0», tú eres un «\*» y los esqueletos son «X». Espero que consigas escapar.

## Cómo funciona el programa

50-90	Llena el array A de espacios vacíos.
110	Establece el número de hoyos permitidos.
120-170	Codifica los símbolos a usar.
180-260	Coloca las vallas en el array.
270-310	Elige posiciones aleatorias para las lápidas.
320-390	Establece las posiciones iniciales para ti y para los esqueletos.
400-420	Coloca los esqueletos en el array.
440-640	Calcula tu nueva posición.
650-740	Mira lo que hay en tu nueva posición. Si no hay nada, te coloca en ella.
770-850	Imprime los distintos mensajes de fin de juego.
860-980	Imprime el cementerio.
1030-110	Coloca un hoyo en el array si tú quieres.
1100-1310	Mueve los esqueletos.

```

●▲ 10 CLS
20 PRINT "EL ENTERRADOR"
30 DIM A(10,20)
40 DIM B(6)
50 FOR I=1 TO 10
60 FOR J=1 TO 20
□ 70 LET A(I,J)=ASC(" ")
80 NEXT J
90 NEXT I
100 LET W=0
110 LET X=5
120 LET Y=ASC("#")
130 LET B=ASC("+")
140 LET C=ASC("0")
150 LET D=ASC(":")
160 LET E=ASC("X")
170 LET Z=ASC(" ")
180 FOR J=1 TO 10
190 LET A(J,1)=D
200 LET A(J,20)=D
210 NEXT J
220 FOR J=1 TO 20
230 LET A(1,J)=D
240 LET A(10,J)=D
250 NEXT J
260 LET A(9,20)=Z
270 FOR J=1 TO 20
□■★ 280 LET F=INT(RND(1)*7+2)
□■★ 290 LET G=INT(RND(1)*15+3)
300 LET A(F,G)=B
310 NEXT J
320 LET M=2
330 LET N=2
340 LET B(2)=19
350 LET B(4)=19
360 LET B(6)=19
370 LET B(1)=4
380 LET B(3)=3
390 LET B(5)=2
400 FOR J=1 TO 5 STEP 2
410 LET A(B(J),B(J+1))=E
420 NEXT J
430 GOSUB 860
440 PRINT
450 PRINT "TECLEA TU MOVIMIENTO ";W
460 PRINT "(PUEDES IR N,S,E,O)"
470 INPUT A$
480 IF A$="N" THEN GOTO 530
490 IF A$="E" THEN GOTO 560
500 IF A$="S" THEN GOTO 600
510 IF A$="O" THEN GOTO 630
520 GOTO 470
530 LET T=N-1
540 LET U=M
550 GOTO 650
560 LET T=N

```



```

570 LET U=M+1
580 IF A(T,U)=Z AND M=19 THEN GOTO 770
590 GOTO 650
600 LET T=N+1
610 LET U=M
620 GOTO 650
630 LET T=N
640 LET U=M-1
650 IF A(T,U)=Z THEN GOTO 710
660 IF A(T,U)=D OR A(T,U)=B THEN GOTO 690
670 IF A(T,U)=C THEN GOTO 800
680 IF A(T,U)=E THEN GOTO 830
690 PRINT "ESE CAMINO ESTA BLOQUEADO"
700 GOTO 440
710 LET A(N,M)=Z
720 LET N=T
730 LET M=U
740 LET A(N,M)=Y
750 GOSUB 1030
760 GOTO 400
770 PRINT "ESTAS LIBRE**"
780 PRINT "TU INDICE DE EFECTIVIDAD
ES ";INT((60-W)/60*(96+X));"%"
790 STOP
800 PRINT "HAS CAIDO EN UNO DE"
810 PRINT "TUS PROPIOS AGUJEROS"
820 STOP
830 PRINT " URK! UN ESQUELETO
TE HA"
840 PRINT "DADO UN SUSTO
DE MUERTE"
850 STOP
860 LET A(N,M)=Y
●▲ 870 CLS
880 LET R=N
890 LET S=M
900 LET W=W+1
910 IF W>60 THEN GOTO 990
920 FOR I=1 TO 10
930 FOR J=1 TO 20
940 PRINT CHR$(A(I,J));
950 NEXT J
960 PRINT
970 NEXT I
980 RETURN
990 PRINT "EL RELOJ INDICA LA"
1000 PRINT "MEDIANOCHE"
1010 PRINT " AGHHHHH!!!!"
1020 STOP
1030 IF X=0 THEN GOTO 1110
1040 PRINT
1050 PRINT " QUIERES CAVAR"
1060 PRINT "UN HOYO (S O N)? "
1070 INPUT B$
1080 IF B$(">")="S" THEN GOTO 1110
1090 LET X=X-1

```


```

1100 LET A(R,S)=C
1110 FOR J=1 TO 5 STEP 2
1120 LET T=B(J)
1130 LET U=B(J+1)
1140 IF A(T,U+1)=Y THEN GOTO 830
1150 IF A(T,U-1)=Y THEN GOTO 830
1160 IF A(T-1,U)=Y THEN GOTO 830
1170 IF A(T+1,U)=Y THEN GOTO 830
1180 IF A$="W" THEN GOTO 1300
1190 IF A$="S" AND A(T+1,U)=Z
THEN GOTO 1270
1200 IF A$="N" AND A(T-1,U)=Z
THEN GOTO 1280
1210 IF A$="E" AND A(T,U+1)=Z
AND M>U THEN GOTO 1240
1220 IF A$="E" AND A(T,U-1)=Z
THEN GOTO 1250
1230 GOTO 1300
1240 LET B(J+1)=B(J+1)+2
1250 LET B(J+1)=B(J+1)-1
1260 GOTO 1290
1270 LET B(J)=B(J)+2
1280 LET B(J)=B(J)-1
1290 LET A(T,U)=Z
1300 NEXT J
1310 RETURN

```

### *Líneas de conversión*

- 10,870 HOME
- ▲ 10,870 PRINT CHR\$(147)
- 70,120-170 CAMBIAR ASC POR CODE
- 280,290 CAMBIAR RND(1) POR RND
- 280,290 CAMBIAR RND(1) POR RND(0)
- ★ 280,290 CAMBIAR RND(1) POR RND(-TIME)



# La casa fantasma

Estás atrapado en una casa encantada donde todo se mueve, incluso las paredes. Si las puertas de salida se alinean, aunque sólo sea por un momento, podrías salir de ella rápidamente. Has encontrado una consola que parece controlar algunos de los movimientos de la casa. Las teclas X y C hacen que la puerta más cercana a ti (la parte de arriba de la pantalla) cambie la dirección. Las teclas N y M tienen el mismo efecto sobre la puerta más lejana a ti. No parece haber ninguna manera de controlar la puerta central.

Mientras luchas frenéticamente con los botones de la consola, oyes pisadas acercándose por el pasillo detrás de ti. El número en el extremo superior izquierdo de la pantalla muestra su avance hacia ti. Si no puedes escapar antes de que las pisadas te alcancen...

...¡Aahhgg...!

Si tu ordenador tiene sonido, intenta añadir efectos para las pisadas y para el final del juego si no puedes escapar.

## Cómo funciona el programa

Define dos arrays. Uno para la posición (P) de cada puerta, el otro para la dirección (S) en que se desplaza.

Define una tabla de estrellas con la anchura (W) de la pantalla y la coloca en LS.

Fija el contador de pisadas a su valor inicial.

Fija la dirección del movimiento de la puerta al principio del juego.

Obtiene posiciones aleatorias de comienzo para las 3 puertas.

Imprime las paredes y las puertas.

Imprime el contador de pisadas en el extremo superior izquierdo.

Comprueba si las pisadas te han alcanzado. Si es así, salta al final del juego.

Comprueba si has ganado.

Obtiene las instrucciones del teclado. Si es necesario, altera la dirección del movimiento cambiando los signos más y menos.

Mueve la puerta central cada 25 pisadas.

Disminuye el valor del contador de pisadas.

Calcula las nuevas posiciones de las puertas. Si alguna alcanza el extremo de la pantalla, el ordenador la mantiene allí.

Vuelve por el siguiente movimiento.

Imprime tu precipitada salida a través de las puertas con su correspondiente mensaje.

Imprime el mensaje cuando has perdido.

```

10 DIM P(3)
20 DIM S(3)
30 CLS
40 PRINT "LA CASA FANTASMA"
50 LET L$=""
60 LET W=31
70 FOR I=1 TO W
80 LET L$=L$+"*"
90 NEXT I
100 LET CT=240
110 LET S(1)=1
120 LET S(3)=-1
130 FOR I=1 TO 3
140 LET P(I)=INT(RND(1)*(W-4)+1)
150 NEXT I
160 CLS
170 FOR I=5 TO 15 STEP 5
180 PRINT TAB(0,I);L$
190 PRINT TAB(P(I/5),I);">"
200 NEXT I
210 PRINT TAB(0,0);CT
220 IF CT<0 THEN GOTO 450
230 IF P(1)=P(2) AND P(2)=P(3) THEN GOTO 390
240 LET Z$=INKEY$
250 IF Z$="" THEN GOTO 300
260 IF Z$="X" THEN LET S(1)=-1
270 IF Z$="C" THEN LET S(1)=1
280 IF Z$="N" THEN LET S(3)=-1
290 IF Z$="M" THEN LET S(3)=1
300 IF CT/25=INT(CT/25) THEN LET P(2)=P(2)+INT(RND(1)*20)-10
310 LET CT=CT-1
320 LET P(1)=P(1)+S(1)
330 LET P(3)=P(3)+S(3)
340 FOR I=1 TO 3
350 IF P(I)<1 THEN LET P(I)=1
360 IF P(I)>(W-4) THEN LET P(I)=W-4
370 NEXT I
380 GOTO 170
390 LET L=P(1)+1
400 FOR I=1 TO 15
410 PRINT TAB(L,I);"M"
420 NEXT I
430 PRINT "ERES LIBRE !!!!"
440 STOP
450 PRINT "DEMASIADO TARDE !"
460 PRINT "LAS PISADAS HAN CESADO ARGHHH!"
470 STOP

```

## Líneas de conversión

### □ Spectrum

```
140 LET P(I)=INT(RND*(W-4)+1)
180 PRINT AT I,0;L#
190 PRINT AT I,P(I/5);" <"
210 PRINT AT 0,0;CT
240 LET Z#=INKEY#
300 IF CT/25=INT(CT/25) THEN
    LET P(2)=P(2)+INT(RND*20)-10
410 PRINT AT I,L;"M"
```

### ▲ C64 y VIC 20

```
30 PRINT CHR$(147)
60 LET W=21
160 PRINT CHR$(147)
170 FOR I=1 TO 3
172 FOR J=1 TO 5
174 PRINT
180 NEXT J
190 PRINT LEFT$(L#,P(I));"
    (";RIGHT$(L#,18-P(I)))
210 PRINT CHR$(19);CT
240 GET Z#
375 PRINT CHR$(19)
395 PRINT CHR$(19)
410 PRINT TAB(L);"M"
```

### ■ MSX

```
140 LET P(I)=INT(RND*(-TIME)*(W-4)+1)
180 LOCATE 0,I:PRINT L#
190 LOCATE P(I/5),I:PRINT ">"-"<"
210 LOCATE 0,0:PRINT CT
300 IF CT/25=INT(CT/25) THEN LET
    P(2)=P(2)+INT(RND*(-TIME)*20)-10
410 LOCATE L,I:PRINT "M"
```

### ● APPLE

```
30 HOME
160 HOME
180 VTAB(I)
195 HTAB(P(I/5))
197 PRINT ">"-"<"
210 VTAB(1):HTAB(1):PRINT CT
240 LET Z#=""
245 IF PEEK(-16384)=127 THEN GET Z#
395 VTAB(1)
410 PRINT TAB(L);"M"
```

### ★ DRAGON

```
140 LET P(I)=RND(30)
170 FOR I=2 TO 12 STEP 5
180 PRINT$(IX32),L#
190 PRINT$(IX32+P((I+3)/5)),") <"
210 PRINT$(CT
240 LET Z#=INKEY#
300 IF CT/25=INT(CT/25) THEN LET
    P(2)=P(2)+RND(20)-10
410 PRINT$(IX32+L),"M"
```

¿Qué ocurre si cambias el número de la línea 100?

¿Y si quitas el signo menos de la línea 120?

Cambia los dos números 25 de la línea 300 por otro número y observa si hay alguna diferencia.

# El laberinto fantasmagórico

En un lugar terrorífico, los pasillos, oscuros e idénticos, no parecen conducir a ninguna parte. Incluso dan la impresión de estar embrujados. Solamente puedes ver de frente. Y sólo puedes moverte en la dirección en que ves. Puedes torcer a la derecha o a la izquierda, pero realmente no irás a ninguna parte, solamente tendrás otra perspectiva. Todo lo que tienes que hacer es encontrar la cruz que marca la salida. Tu posición se representa por una «Y» y las paredes con «=».

¡Glup! Está encantado. Los fantasmas se representan con la letra «F». Si te acercas demasiado a uno, serás trasladado a cualquier otra parte del laberinto, sin saber dónde estás o hacia dónde te diriges. Éstas son las teclas que puedes usar: «X» para avanzar, «N» para torcer 90 grados a la izquierda, «M» para girar 90 grados a la derecha.

## Cómo funciona el programa

10 DIM E(70)	Define un área de memoria (array) para el laberinto.
20 DIM V\$(4,3)	Define una mini-matriz dentro del laberinto, que es la parte de él que puedes ver en la pantalla.
30 DIM F(3)	Establece el espacio para los datos que llenan una fila de esta mini-matriz.
40 LET W\$=""	
50 LET W\$=W\$+"0000000000"	
60 LET W\$=W\$+"01111100110"	
70 LET W\$=W\$+"0010011100"	
80 LET W\$=W\$+"0011010110"	
90 LET W\$=W\$+"0110100100"	
100 LET W\$=W\$+"0011111100"	
110 LET W\$=W\$+"0000009000"	Contiene los datos para el laberinto. 1=pasillo; 0=pared; 9=salida.
120 FOR I=1 TO 70	
130 LET E(I)=VAL(MID\$(W\$,I,1))	Bucle que coloca los datos en el array E.
140 NEXT I	
150 LET S=-1	Fija el número de veces que el fantasma se ha movido a -1 al comienzo.
160 LET G=12	Inicializa el número de fantasmas a cero.
170 LET X=INT(RND(1)*50)+10	Elige una posición de comienzo aleatoria para ti.
180 IF E(X) <> 1 THEN GOTO 170	Si esta posición no está en un pasillo, vuelve por otra.
190 GOSUB 860	Va a la subrutina que sitúa al fantasma.
200 LET D=INT(RND(1)*4)+1	Elige tu dirección de avance al comienzo del juego.
210 IF X=0+10 OR X=0-10 THEN GOTO 170	Comprueba si estás junto a un fantasma y si es así te lleva a un nuevo lugar.
220 IF X=0+1 OR X=0-1 THEN GOTO 170	
230 LET H=H+1	Incrementa el número de movimientos que has hecho.
240 IF H=5 THEN GOSUB 860	Va a la subrutina de movimiento del fantasma cada 5 movimientos.
250 GOSUB 390	Salta a la subrutina que imprime la sección del laberinto que puedes ver.
260 LET A\$=INKEY\$	Interpreta tus instrucciones. Cambia el valor de la dirección, D, si es necesario.
270 IF A\$="" THEN GOTO 260	
280 IF A\$="N" THEN LET D=D-1	
290 IF A\$="M" THEN LET D=D+1	
300 IF D=5 THEN LET D=1	
310 IF D=0 THEN LET D=4	
320 IF A\$="X" THEN GOTO 210	Regresa para hacer el siguiente movimiento si no has avanzado.
330 IF D=1 AND E(X-10) <> 0 THEN LET X=X-10	
340 IF D=3 AND E(X+10) <> 0 THEN LET X=X+10	Comprueba si te puedes mover.
350 IF D=2 AND E(X+1) <> 0 THEN LET X=X+1	
360 IF D=4 AND E(X-1) <> 0 THEN LET X=X-1	
370 IF E(X)=9 THEN GOTO 930	Comprueba si has alcanzado la salida.
380 GOTO 210	Si no es así, regresa para hacer otro movimiento.



## Añade sonidos

Si tu ordenador tiene sonidos, intenta hacer un ruido cada vez que aparezca un fantasma.

```

390 FOR I=1 TO 4
400 LET T=I-1
410 ON D GOTO 420,460,500,540
420 LET F(1)=X-10*T+1
430 LET F(2)=X-10*T
440 LET F(3)=X-10*T-1
450 GOTO 570
460 LET F(1)=X+10*T
470 LET F(2)=X+T
480 LET F(3)=X-10*T
490 GOTO 570
500 LET F(1)=X+10*T-1
510 LET F(2)=X+10*T
520 LET F(3)=X+10*T+1
530 GOTO 570
540 LET F(1)=X-T-10
550 LET F(2)=X-T
560 LET F(3)=X-T+10
570 FOR J=1 TO 3
580 IF F(J)=1 OR F(J)=69 THEN GOTO 840
590 IF E(F(J))=0 THEN LET V*(I,J)="*
600 IF E(F(J))=1 THEN LET V*(I,J)=" "
610 IF E(F(J))=9 THEN LET V*(I,J)="+"
620 IF E(F(J))=2 THEN LET V*(I,J)="F"
630 NEXT J
640 NEXT I
650 LET V*(1,2)="Y"
660 CLS
670 PRINT
680 PRINT "*** LABERINTO FANTASHAGORICO ***"
690 PRINT
700 PRINT "AVANZAR" X*
710 PRINT "DERECHA" H*
720 PRINT "IZQUIERDA" N*
730 FOR I=1 TO 5
740 PRINT
750 NEXT I
760 FOR I=4 TO 1 STEP -1
770 LET P*=""
780 FOR J=3 TO 1 STEP -1
790 LET P*=P*+V*(I,J)
800 NEXT J
810 PRINT TAB(7);P*
820 NEXT I
830 RETURN
840 LET V*(I,J)="*
850 GOTO 630
860 LET E(G)=1
870 LET G=INT(RND(1)*50)+10
880 IF E(G)<1 THEN GOTO 870
890 LET E(G)=2
900 LET H=0
910 LET S=S+1
920 RETURN
930 PRINT "HAS ESCAPADO"
940 PRINT "EN "S*5+H;" MOVIMIENTOS"
950 STOP

```

## Líneas de conversión

```

■ 5 CLEAR 200
130 LET E(I)=VAL(W*(I TO I))
170 LET X=INT(RND*50)+10
170 LET X=INT(RND(0)*50)+10
★ 170 LET X=INT(RND(-TIME)*50)+10
200 LET D=INT(RND*4)+1
■ 200 LET D=INT(RND(0)*4)+1
★ 200 LET D=INT(RND(-TIME)*4)+1
● 255 LET A*=""
▲ 260 GET A*
● 260 IF PEEK(-16384)>127 THEN GET A*
410 GOTO 380+40*D
▲ 660 PRINT CHR*(147)
● 660 HOME
870 LET G=INT(RND*50)+10
■ 870 LET G=INT(RND(0)*50)+10
★ 870 LET G=INT(RND(-TIME)*50)+10

```

## Para mejorar los gráficos

Intenta sustituir los símbolos «+», «F», «Y» y «#» por caracteres gráficos (y también colores). Esto implicará cambios en las líneas 590-620 y 840.

Subrutina que calcula:

- a) qué cuadrados puedes ver desde donde estás.
- b) qué caracteres poner en esos cuadrados.
- y c) imprime instrucciones y la sección de laberinto que puedes ver.

## Amplía el laberinto

Este laberinto está basado en una matriz de 70 cuadrados. Para hacerla más grande, has de cambiar las líneas siguientes, sustituyendo N por el número de cuadrados que quieras que tenga tu nuevo laberinto. (N debe ser múltiplo de 10.):

10: Sustituye 70 por tu número, N.  
50-110: Añade líneas extra de unos y ceros en 55, 65, etc., de tal manera que el número total de caracteres de la matriz sea N.

120: Cambia 70 por N.

170 y 870: Sustituye 50 por N-20.

580: Cambia 69 por N-1.

Subrutina para desplazar el fantasma cada 5 movimientos.

Mensaje de fin del juego.

# Espiritismo

Llegan los mensajes de los espíritus, letra a letra. Quieren que recuerdes las letras y las introduzcas en el ordenador en el orden correcto. Si cometes errores, se enfadarán mucho. ¡Mucho!

Mira las estrellas de tu pantalla: te mostrarán las letras de los mensajes de los espíritus.

```

10 LET S=0
20 LET G=0
30 LET CS=64
40 CLS
50 PRINT
60 PRINT TAB(8);"ESPIRITISMO"
70 FOR I=1 TO 8
80 LET X=6+I
90 LET Y=5
100 LET A$=CHR$(CS+I)
110 GOSUB 710
120 LET Y=Y+1
130 LET A$=CHR$(CS+22-I)
140 GOSUB 710
150 NEXT I
160 FOR I=1 TO 5
170 LET X=X+1
180 LET Y=Y+1
190 LET A$=CHR$(CS+27-I)
200 GOSUB 710
210 LET X=X+1
220 LET A$=CHR$(CS+8+I)
230 GOSUB 710
240 NEXT I
250 LET P$=""
260 LET N=INT(RND(1)*4+3)
270 FOR I=1 TO N
280 LET A$=""
290 LET L=INT(RND(1)*26+1)
300 LET S$=CHR$(CS+L)
310 LET P$=P$+S$
320 LET D=4
330 IF L<22 THEN LET D=3
340 IF L<14 THEN LET D=2
350 IF L<9 THEN LET D=1
360 ON D GOTO 370,400,430,460
370 LET Y=Y+6
380 LET X=X+6
390 GOTO 480
400 LET X=X+15
410 LET Y=Y+3
420 GOTO 480
430 LET Y=Y+10
440 LET X=X+28-L
450 GOTO 480
460 LET X=X+6
470 LET Y=Y+32-L
480 GOSUB 710

```

La línea 490 controla a qué velocidad aparecen las estrellas en la pantalla. Cámbiala poniendo un número más bajo para aumentar la velocidad del juego.

```

490 FOR T=1 TO 600
500 NEXT T
510 LET A$=""
520 GOSUB 710
530 NEXT I
540 LET A$=""
550 LET X=0
560 LET Y=13
570 GOSUB 710
580 INPUT R$
590 IF R$=P$ THEN GOTO 670
600 LET G=G+1
610 IF G=1 THEN PRINT "LA MESA
    ENPIEZA A VIBRAR"
620 IF G=2 THEN PRINT "LA
    LUZ PARPADEA"
630 IF G=3 THEN GOTO 730
640 FOR T=1 TO 1000
650 NEXT T
660 GOTO 40
670 LET S=S+N
680 IF S<50 THEN GOTO 40
690 PRINT "LOS ESPIRITUS SE HAN IDO"
700 STOP
710 PRINT TAB(X,Y);A$;
720 RETURN
730 PRINT "UN PAR DE GELIDAS MANOS"
740 PRINT "ESTRECHA TU CUELLO"
750 STOP

```

## Cómo funciona el programa

Después de que hayas tecleado el programa y jugado unas cuantas veces, intenta averiguar qué función tiene cada una de las líneas del programa. (Nota: D es una variable que dice al ordenador en qué lado del cuadrado —1, 2, 3 ó 4— está la siguiente letra del mensaje.)

## Líneas de conversión

```

40 PRINT CHR$(147)
40 HOME
260 LET N=INT(RND*4+3)
260 LET N=INT(RND(0)*4+3)
260 LET N=INT(RND(-TIME)*4+3)
290 LET L=INT(RND*26+1)
290 LET L=INT(RND(0)*26+1)
290 LET L=INT(RND(-TIME)*26+1)
360 GOTO 340+30*D
490 FOR T=1 TO 100
640 FOR T=1 TO 300
710 PRINT AT Y,X;A$;
710 PRINT CHR$(19);FOR K=1
    TO Y:PRINT:NEXT:PRINT TAB(X);A$;
710 PRINT@32*Y+X,A$;
710 VTAB(Y):HTAB(X):PRINT A$;
710 LOCATE X,Y:PRINT A$;

```

# Respuestas a las cuestiones planteadas

Aquí hay algunas soluciones para las cuestiones propuestas en este libro. Puede ocurrir que tus respuestas sean diferentes; pero si funcionan en tu ordenador, valen. Comprueba, de todos modos, que tus soluciones son tan limpias y directas como las que se dan aquí.

## Pesadilla informática

```

● 90 LET N#=CHR$(INT(RND(1)*26+65))
  100 PRINT TAB(5);N#

220 IF F#(>)N# THEN GOTO 240

230 LET S=S+10+ASC(N#)-65

90 LET N#=CHR$(INT(RND*26+65))
■ 90 LET N#=CHR$(INT(RND(0)*26+65))
★ 90 LET N#=CHR$(INT(RND(-TIME)*26+65))
230 LET S=S+10+CODE(N#)-65
  
```

Elige una letra y la imprime en la pantalla, en lugar de un número.

Comprueba si la tecla pulsada y la letra elegida son la misma.

Incrementa el tanteo. (Depende de la letra: por ejemplo, con Z obtienes más puntos que con A.)

Ten en cuenta las diferentes versiones para los ordenadores.

## Los números mágicos

```

5 LET L=0
6 W=0

355 LET L=L+1
360 GOTO 390
380 LET W=W+1
390 PRINT
400 PRINT "EL HECHICERO    TU"
410 PRINT
420 PRINT "    *ILI"    "IW"
430 FOR Q=1 TO 1000
440 NEXT Q
450 GOTO 30
  
```

Inicializa el número de partidas ganadas y perdidas al principio del juego.

Incrementa el número de veces que has perdido.

Incrementa el número de veces que has ganado.

Imprime el marcador.

Pausa para que puedas leer el tanteo.

Cambia este número para los ordenadores más rápidos.

Regresa para jugar otra vez.

## La mujer-araña

```

5 LET NG=0

355 LET NG=NG+1

360 IF G#=T# THEN GOTO 370

362 PRINT "NO, ESTAS EQUIVOCADO"

364 LET G=G+5

366 IF NG=2 THEN GOTO 410

368 GOTO 230
  
```

Inicializa el número de intentos.

Incrementa el número de intentos.

Si la letra es correcta, para el juego.

Imprime el mensaje que dice que no es correcta la letra.

Incrementa el número de intentos en 5, como penalización por haberte equivocado.




Si has usado dos intentos, termina el juego.

Regresa para jugar la siguiente baza. (Con una pausa para que puedas leer el mensaje de la pantalla.)

# JUEGOS DE ORDENADOR

Cada uno de estos nuevos libros contiene sencillos programas de juegos para utilizar en los ordenadores domésticos de mayor difusión. Junto a cada programa hay explicaciones sobre cómo funciona, así como problemas y sugerencias para introducir cambios que compliquen las aventuras o las hagan más sencillas, según los deseos del usuario. Por medio de estos apasionantes juegos, incluso los que nunca hayan utilizado un ordenador aprenderán con facilidad cómo funciona un programa sencillo, y podrán diseñar alguno por sí mismos. Al final de cada libro aparecen las claves de conversión para diversos ordenadores, así como un glosario de los términos más usuales en BASIC.

## Títulos de la serie



**EDICIONES GENERALES ANAYA**